
OR – Organizacija računalnikov 1

Fakulteta za računalništvo in informatiko

STM32H7

Vhodno / izhodne naprave

Prekinitve

SysTick Časovnik

= Mini RTOS

Prekinitve – Zakaj ?

Prekinitve - dogodki
Optimizacija programske opreme za 

ustrezne reakcije na dogodke v 

realnem času:
- dogodkovno vodene reakcije

- ciklične prekinitve s časovniki

Prožijo se ustrezni Prekinitveno servisni 

podprogrami – PSP (ang. ISR)

Veliko virov (naprav) povezanik na krmilnik (NVIC), ki naprej sporoča zahteve na CPE

http://www.fri.uni-lj.si/si

Prekinitve - periodične
Izvedba periodičnih opravil, kot je 

preklop med neodvisnimi opravili:
- Shranimo stanje trenutnega 

opravila

- Vzpostavimo stanje naslednjega 

opravila

- Izvajamo do naslednje prekinitve

Periodično se prožijo ustrezni 

Prekinitveno servisni podprogrami – PSP 

(ang. ISR) – npr. SysTick_Handler

main()

ISR()

SysTick

time

CPU

Task #0

Task #1

Context
switch

http://www.fri.uni-lj.si/si

STM32H750XB

http://www.fri.uni-lj.si/si


OR – Organizacija računalnikov 6

Fakulteta za računalništvo in informatiko

Priključitev :

• Mikro USB priklop na daljši stranici (nad LCD, srednji !!!)

Poseben začetni projekt (github) in info za STM32H7 (e-učilnica):

• dodajanje vsebine (Main.s):

Delo na STM32H7 razvojnem sistemu

http://www.fri.uni-lj.si/si

Časovniki

SysTick

http://www.fri.uni-lj.si/si

Prekinitve – STM32
Izvedba periodičnih opravil, kot je preklop med neodvisnimi opravili:

Program

time

ISR 

Stacking Unstacking
xxxxxxxx SP + 0x20
xPSR SP + 0x1C

PC(r15) SP + 0x18
LR(r14) SP + 0x14
r12 SP + 0x10
r3 SP + 0x0C
r2 SP + 0x08
r1 SP + 0x04
r0 SPSP

Full 

Descending 

StackBX LR

http://www.fri.uni-lj.si/si

Vira: Reference & Programming manuals

Ponovitev OR-LAB 9: SysTick+prek.

http://www.fri.uni-lj.si/si

SysTick časovnik + prek. – stanje , nastavitve

Osnovni registri za delovanje SysTick časovnika:

SYST_CSR :  vklop časovnika 

CLKSOURCE=1, TICKINT=1, ENABLE=1

COUNTERFLAG=1, ko prešteje do 0 (postavi na SYST_RVR in nadaljuje)

SYST_RVR : zač. vrednost štetja (šteje proti 0)

SYST_RVR = število period 

SYST_CVR : trenutna vrednost števca 

SYST_CVR = nekje med SYST_RVR in 0

Bazni naslov za registre SysTick je 0xE000E010 Ponovitev OR-LAB 9: SysTick+prek.

http://www.fri.uni-lj.si/si

SysTick časovnik (Registri za nastavitve delovanja)

Vklop prekinitve

Ponovitev OR-LAB 9: SysTick+prek.

http://www.fri.uni-lj.si/si

Potrebni koraki za krmiljenje časovnika SysTick:

1. SYST_RVR (Reload Value Register): Value SYSTICK_RELOAD_1MS

2. SYST_CVR (Current Value Register): 0, reset to zero 

3. SYST_CSR (Control/Status Register): 0b111 : Proc. Clock, TickInt, Enable

-> Start SysTick

4. Delovanje: 

Proženje SysTick_Handler vsako 1 ms

SysTick_Handler :
.global SysTick_Handler

.section .text.SysTick_Handler,"ax",%progbits

.type SysTick_Handler, %function

SysTick_Handler:

push {r3, r4, r5, lr}

…

RET: pop {r3, r4, r5, pc}

SysTick Časovnik – krmiljenje 
Ponovitev OR-LAB 9: SysTick+prek.

http://www.fri.uni-lj.si/si

Prekinitveni vektorji

// Start of text section
.section .text
///////////////////////////////////////////////////////////////////////////////
// Vectors
///////////////////////////////////////////////////////////////////////////////
// Vector table start
// Add all other processor specific exceptions/interrupts in order here
.long    __StackTop                 // Top of the stack. from linker script
.long    _start +1                  // reset location, +1 for thumb mode
.word NMI_Handler
.word HardFault_Handler
.word MemManage_Handler
.word BusFault_Handler
.word UsageFault_Handler
.word 0
.word 0
.word 0
.word 0
.word SVC_Handler
.word DebugMon_Handler
.word 0
.word PendSV_Handler
.word SysTick_Handler

/* External Interrupts */
  .word     WWDG_IRQHandler                   /* Window WatchDog              */
  .word     PVD_IRQHandler                    /* PVD through EXTI Line detection */
  .word     TAMP_STAMP_IRQHandler             /* Tamper and TimeStamps through the EXTI line */
  .word     RTC_WKUP_IRQHandler               /* RTC Wakeup through the EXTI line */

.global SysTick_Handler

.section .text.SysTick_Handler,"ax",%progbits

.type SysTick_Handler, %function

SysTick_Handler:

push {r3, r4, r5, r6, lr}
…

Ponovitev OR-LAB 9: SysTick+prek.

http://www.fri.uni-lj.si/si

Prekinitve (Register za nastavitve vektorske tabele v RAM pomnilniku)

.equ VTOR,0xE000ED08

RELLOC_VECTBL:
push {r0, r1, lr}
ldr r1, =VTOR // Set Vector table addr. to 0x24000000
ldr r0, =0x24000000
str r0, [r1]
pop {r0, r1, pc}

Dodamo samo, če poganjamo 
kodo iz RAM pomnilnika !!!
Sicer je naslov vektorske 
tabele 0x08000000 (Flash)

Ponovitev OR-LAB 9: SysTick+prek.

http://www.fri.uni-lj.si/si

Začetna koda:

main:
#ifdef RAM_LinkScript
// Relocating Vector table to RAM (only for RAM Linker Script)

bl RELLOC_VECTBL
#endif

bl INIT_GPIOs
bl INIT_USART3 // Priprava USART3 naprave
bl ITM_Init

// MiniRTOS_Init
// Set new User SP that will be used for user program (MSP is reset default)

ldr r0,=USERSTACK_End-4
msr psp, r0

// Set PSP as SP when stacking/unstacking user processes
// Access special registers

mrs r0, CONTROL // Read CONTROL into R0
orr r0, R0, #0x2 // Set SPSEL (PSP to be used in Thread mode) to select PSP
msr CONTROL, r0 // Write R0 into CONTROL
isb // Instruction Synchronization Barrier - needed

CPSID I // Disable interrupts

bl INIT_TC_PSP // Initialize SysTick periodic interrupt (every msec)

CPSIE I // Enable interrupts (context switches will happen)

b TASK0_Start // Go into first task (it will switch with TASK1 every msec)

__end: b __end

SysTick Časovnik s prekinitvami – Mini RTOS 

Števec v r8 šteje 500 ms

Zastavica v r7 pove stanje LED diod

Vse se dogaja v PSP !

http://www.fri.uni-lj.si/si

SysTick_Handler :

SysTick_Handler:
/// STEP 1 - SAVE THE CURRENT TASK CONTEXT
/// Disable interrupts
CPSID I

/// Push remaining registers R4 to R11
// push {r4-r11}
mrs r3, psp // move PSP to r3
stmfd r3!, {r4-r11} // Push onto a FD Stack

ldr r0,=ACT_TASKID // Check current Active task ID
ldr r1,[r0]
cmp r1,#0
bne CONT1

// TASK 0 current
mov r2,#1 // Task1 is new
str r2,[r0]

ldr r0,=StackPointer0 // Store current PSP
str r3,[r0]

ldr r0,=PCB1
ldr r1,=StackPointer1
b STEP2

…

SysTick MiniRTOS – PSP -> Context Switch

Števec v r8 šteje 500 ms

Zastavica v r7 pove stanje LED diod

Vse se dogaja v PSP !

…

// TASK 1 current
CONT1:
mov r2,#0 // Task0 is new
str r2,[r0]

ldr r0,=StackPointer1 // Store current PSP
str r3,[r0]

ldr r0,=PCB0
ldr r1,=StackPointer0

/// STEP 2: LOAD THE NEW PSP and TASK CONTEXT FROM ITS 
STACK TO THE CPU REGISTERS.
STEP2:
ldr r3,[r1] // Read new PSP into r3

/// Pop registers R4-R11
ldmfd r3!, {r4-r11} // Load from a FD Stack

msr psp, r3 // move r3 to PSP for upcoming task

/// Enable interrupts
CPSIE I

BX LR

http://www.fri.uni-lj.si/si

TASK0_Start:

bl LED1_ON
mov r0,#0
mov r1,#'1'
bl ITM_Send

mov r0,#'0'
bl SEND_UART

mov r0,#500
bl DELAY // Zakasnitev SW Delay: r0 x 1msec

bl LED1_OFF
mov r0,#0
mov r1,#'0'
bl ITM_Send

mov r0,#500
bl DELAY // Zakasnitev SW Delay: r0 x 1msec

b TASK0_Start

SysTick MiniRTOS – Task0, Task1 

Števec v r8 šteje 500 ms

Zastavica v r7 pove stanje LED diod

Vse se dogaja v PSP !

TASK1_Start:

bl LED2_OFF
mov r0,#1
mov r1,#'+'
bl ITM_Send

mov r0,#'1'
bl SEND_UART

mov r0,#500
bl DELAY // Zakasnitev SW Delay: r0 x 1msec

bl LED2_ON
mov r0,#1
mov r1,#'-'
bl ITM_Send

mov r0,#500
bl DELAY // Zakasnitev SW Delay: r0 x 1msec

b TASK1_Start 

.equ STACKSIZE , 240 // + 16 registers of PCB

USERSTACK_Start: .space (STACKSIZE+16) * 4
USERSTACK_End:

StackPointer0: .word PCB0 // PSP for Task0
StackPointer1: .word PCB1 // PSP for Task1

ACT_TASKID: .word 0 // current process ID (0,1,...)

STACK1_Start: .space STACKSIZE * 4
// PCB: ISR saves: R4-R11 (8regs), CPU saves: R0-R3, R12, LR, PC, PSR
PCB1: .word 4,5,6,7,8,9,10,11, 0,1,2,3, 12, 0xFFFFFFFF, TASK1_Start, 0x01000000
STACK1_End: // was PSP Thumb mode

http://www.fri.uni-lj.si/si

TASK0_Start:

…

b TASK0_Start

SysTick MiniRTOS – Task0, Task1 

Števec v r8 šteje 500 ms

Zastavica v r7 pove stanje LED diod

Vse se dogaja v PSP !

TASK1_Start:

…

b TASK1_Start 

.equ STACKSIZE , 240 // + 16 registers of PCB

USERSTACK_Start: .space (STACKSIZE+16) * 4
USERSTACK_End:

StackPointer0: .word PCB0 // PSP for Task0

ACT_TASKID: .word 0 // current process ID (0,1,...)

STACK1_Start: .space STACKSIZE * 4
// PCB: ISR saves: R4-R11 (8regs), CPU saves: R0-R3, 
R12, LR, PC, PSR
PCB1: .word 4,5,6,7,8,9,10,11, 0,1,2,3, 12, 
0xFFFFFFFF, TASK1_Start, 0x01000000
STACK1_End: // was PSP Thumb mode

StackPointer1: .word PCB1 // PSP for Task1

r4-r11: SysTick Handler 

dodatno shrani na sklad

mrs r3, psp // move PSP to r3
stmfd r3!, {r4-r11} // Push onto a FD Stack

Kazalca na sklad sta shranjena posebej:

StackPointer0, StackPointer1

http://www.fri.uni-lj.si/si

Stack Pointer: MSP vs PSP

19

• MSP: Main Stack Pointer (selected at reset)

• PSP:  Process Stack Pointer

• In assembly, both MSP and PSP are called R13 or SP

For interrupts, which stack 

does auto 

stacking/unstacking use?

Depends on 

• processor mode: thread vs handler 

• setting in the control register

http://www.fri.uni-lj.si/si

SysTick Interrupt

20

http://www.fri.uni-lj.si/si

Control Register

21

31 - 3 2 1 0

Reserved FPCA SPSEL nPRI
V

0:  FP inactive (default)

1:  FP active

0:  SP = MSP (default)

1:  SP = PSP if in Thread Mode

0:  Thread mode has privileged access (default)

1:  Thread mode has unprivileged access

By default, MSP is used.

// MiniRTOS_Init

// Set new User SP that will be used for user program (MSP 
is reset default)
ldr r0,=USERSTACK_End-4
msr psp, r0

// Set PSP as SP when stacking/unstacking user processes
// Access special registers
mrs r0, CONTROL // Read CONTROL into R0
orr r0, R0, #0x2 // Set SPSEL (PSP to be used in Thread
mode) to select PSP
msr CONTROL, r0 // Write R0 into CONTROL
isb // Instruction Synchronization Barrier - needed

http://www.fri.uni-lj.si/si

CONTROL register

http://www.fri.uni-lj.si/si

ITM Trace - Init

ITM_Init:
push {r0, r1, lr}

ldr r0, =M7_ITM_BASE // Base address
mov r1, #0b11
str r1,[r0,#M7_ITM_TER]

mov r1, #1
str r1,[r0,#M7_ITM_TCR]

pop {r0, r1, pc}

http://www.fri.uni-lj.si/si

ITM Trace - Send
ITM_Send:
push {r2, lr}

ldr r2, =M7_ITM_BASE // Base address

cmp r0,#0 // Channel 0 ?
bne Ch1

str r1,[r2] // Channel 0
b contx

Ch1:
str r1,[r2,#4] // Channel 1

contx:
pop {r2, pc}

http://www.fri.uni-lj.si/si

ITM Registers

http://www.fri.uni-lj.si/si

CubeIDE – SFR okno

http://www.fri.uni-lj.si/si

CubeIDE – SFR okno

http://www.fri.uni-lj.si/si

CubeIDE – Sledenje delovanja

http://www.fri.uni-lj.si/si

